Abstract
The SAS® Macro Language is a powerful tool for extending the capabilities of the SAS System. This hands-on workshop teaches essential macro coding concepts, techniques, tips and tricks to help beginning users learn the basics of how the Macro language works. Using a collection of proven Macro Language coding techniques, attendees learn how to write and process macro statements and parameters; replace text strings with macro (symbolic) variables; generate SAS code using macro techniques; manipulate macro variable values with macro functions; create and use global and local macro variables; construct simple arithmetic and logical expressions; interface the macro language with the SQL procedure; store and reuse macros; troubleshoot and debug macros; and develop efficient and portable macro language code.

Introduction
The Macro Language serves as an extension to the SAS System for the purpose of generating text in the form of SAS code, including partial and/or complete statements, DATA steps, PROC steps, variables, text strings, functions, informats, formats, expressions, comparison and logical operators, and other elements related to SAS syntax. As a language, the macro language provides users with its own set of statements, options, functions, as well as its own compiler.

One essential difference between macro code and SAS code is that macro code is compiled and executed before SAS DATA step and PROC step code, and the generated text is then processed by the SAS System. When programming with macro statements, the resulting program is called a MACRO. The Macro Language has its own rules for using the various statements and parameters. The Macro environment can be thought of as a lower level (3rd Generation) programming environment within the SAS System.

Macro Language Basics
The macro language provides an additional set of tools to: 1) communicate between SAS steps, 2) construct executable and reusable code, 3) design custom languages, 4) develop user-friendly routines, and 5) conditionally execute DATA or PROC steps.

When a program is run, the SAS System first checks to see if a macro statement exists. If the program does not contain any macro statements, then processing continues as normal with the DATA or PROC step processor. If the program does contain one or more macro statements, then the macro processor must first execute them. The result of this execution is the production of character information, macro variables, or SAS statements, which are then be passed to the DATA or PROC step processor. The control flow of a macro process appears in Figure 1 below.

Figure 1. Macro Program Control Flow.
The SAS System Log displays information about the compilation and execution of a SAS program. This information is a vital part of any SAS execution which when viewed provides information about: 1) What statements were executed, 2) What SAS System data sets were created, 3) The number of variables and observations each data set contains, and 4) The time and memory expended by each DATA and PROC step.

The Anatomy of a Macro
Every macro begins with a %MACRO and must contain a name for the macro. To close a macro, a %MEND is used and can optionally specify the macro name for documentation reasons. Macro text can include any of the following information:

- Constant Text
- Macro Variables
- Macro Functions
- Macro Program Statements
- Macro Expressions

**Constant Text**
The macro language treats constant text as character strings. Examples include:

- SAS Data Set Names
- SAS Variable Names
- SAS Statements

**Macro Variables**
Macro variables (symbolic variables) are not DATA step variables, but belong to the SAS System macro language. Symbolic variables, once defined, can take on many different values during the execution of a macro program. Basic rules that apply to the naming of symbolic variables are:

- A name can be one to eight characters in length
- A name must begin with a character (A-Z) or underscore (_)
- Letters, numbers, and underscores can follow the first character

Basic rules that apply to the use of symbolic variables include:

- Values range from 0 to 65,534 characters in length
- The number of characters assigned to a macro variable determines its length – no length declaration is made
- Leading and trailing blanks are not stored with the value
- May be referenced (called) inside or outside of a macro by immediately prefixing an ampersand (&) before the name
- The macro processor replaces (substitutes) the symbolic variable with the value of the symbolic variable

A couple examples are provided to help clarify the creation and use of macro variables.
References Inside a Macro:

```plaintext
%LET NAME=USERFILE.MASTER;
%MACRO M;
   PROC MEANS DATA=&NAME;
   RUN;
%MEND M;
```

References Outside a Macro:

```plaintext
PROC PRINT DATA=&NAME;
RUN;
```

**Macro Functions**

Macro functions are available to process text in macros and with macro variable values. Some macro functions are associated with DATA step functions while others are used only in the macro processor. You may notice a similarity between DATA step functions and macro functions. To illustrate how macro functions can be used, a few examples are shown below.

**Examples:**

```plaintext
%INDEX(argument1,argument2)

%STR(argument)

%UPCASE(argument)

%BQUOTE(argument)
```

**Macro Program Statements**

The macro language provides a powerful language environment for users to construct and use macro programs. There are a number of Macro program statements, many of which resemble DATA step statements in use and functionality. Macro program statements are available to instruct the macro processor what to do. Each statement begins with a percent sign (%) and is terminated with a semi-colon (;). The statements are executed by the macro processor and then passed to either the DATA or PROC step for processing.

**Examples:**

```plaintext
%DO;
%END;

%GLOBAL macro-variable;

%MACRO name[(parameters)/STMT];
```

**Macro Expressions**

Macro expressions consist of macro statements, macro variable names, constant text, and/or function names combined together. Their purpose is to tie processing operations together through the use of operators and parentheses.

**Examples:**

```plaintext
IF &TOTAL > 999 THEN WEIGHT=WEIGHT+1;

&CHAR = %LENGTH(&SPAN)

&COUNT = %EVAL(&COUNT + 1);
```
**Tip #1 – Debugging a Macro with SAS System Options**

The SAS System offers users a number of useful system options to help debug macro issues and problems. The results associated with using macro options are automatically displayed on the SAS Log. Specific options related to macro debugging appear in alphabetical order in the following table.

<table>
<thead>
<tr>
<th>SAS Option</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>MACRO</td>
<td>Specifies that the macro language SYMGET and SYMPUT functions be available.</td>
</tr>
<tr>
<td>MEMERR</td>
<td>Controls Diagnostics.</td>
</tr>
<tr>
<td>MEMRPT</td>
<td>Specifies that memory usage statistics be displayed on the SAS Log.</td>
</tr>
<tr>
<td>MERROR</td>
<td>Presents Warning Messages when there are misspellings or when an undefined macro is called.</td>
</tr>
<tr>
<td>MLOGIC</td>
<td>Macro execution is traced and displayed on the SAS Log for debugging purposes.</td>
</tr>
<tr>
<td>MPRINT</td>
<td>SAS statements generated by macro execution are traced on the SAS Log for debugging purposes.</td>
</tr>
<tr>
<td>SYMBOLGEN</td>
<td>Displays text from expanding macro variables to the SAS Log.</td>
</tr>
</tbody>
</table>

**Tip #2 – Using the Autocall Facility to Call a Macro**

Macro programs can be stored as SAS programs in a location in your operating environment and called on-demand using the built-in autocall facility. Macro programs stored this way are defined once, and referenced (or called) anytime needed. This provides an effective way to store and manage your macro programs in a library aggregate. To facilitate the autocall environment, you will need to specify the SAS System options presented in the following table.

<table>
<thead>
<tr>
<th>SAS Option</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>MAUTOSOURCE</td>
<td>Turns on the Autocall Facility so stored macro programs are included in the search for macro definitions.</td>
</tr>
<tr>
<td>MRECALL</td>
<td>Turns on the capability to search stored macro programs when a macro is not found.</td>
</tr>
<tr>
<td>SASAUTOS=</td>
<td>Specifies the location of the stored macro programs.</td>
</tr>
</tbody>
</table>

**Tip #3 – Accessing the SAS Institute-supplied Autocall Macros**

Users may be unaware that SAS Institute has provided as part of your SAS software an autocall library of existing macros. These autocall macros are automatically found in your default SASAUTOS fileref. For example, the default location of the SASAUTOS fileref under Windows XP Professional on my computer is c:\program files\sas\sas 9.1\core\sasmacro. Readers are encouraged to refer to the SAS Companion manual for the operating environment you are running under for further details.

Numerous SAS-supplied autocall macros are included – many of which act and behave as macro functions. It is worth mentioning that these autocall macros provide a wealth of effective coding techniques and can be useful as a means of improving macro coding prowess in particular for those users who learn by example. The following table depicts an alphabetical sampling of the SAS Institute-supplied autocall macros for SAS 9.1.

<table>
<thead>
<tr>
<th>SASAUTOS Macro Name</th>
<th>SASAUTOS Macro Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>%CHNGCASE</td>
<td>This macro is used in the change dialog box for pmenus.</td>
</tr>
<tr>
<td>%CMPRES</td>
<td>This macro returns the argument passed to it in an unquoted form with multiple blanks compressed to single blanks and also with leading and trailing blanks removed.</td>
</tr>
<tr>
<td>%DATATYP</td>
<td>The DATATYP macro determines if the input parameter is NUMERIC or CHARacter data, and returns either CHAR or NUMERIC depending on the value passed through the parameter.</td>
</tr>
<tr>
<td>%LEFT</td>
<td>This macro returns the argument passed to it without any leading blanks in an unquoted form.</td>
</tr>
<tr>
<td>%LOWCASE</td>
<td>This macro returns the argument passed to it unchanged except that all upper-case alphabetic characters are changed to their lower-case equivalents.</td>
</tr>
<tr>
<td>%SYSRC</td>
<td>This macro returns a numeric value corresponding to the mnemonic string passed to it and should only be used to check return code values from SCL functions.</td>
</tr>
<tr>
<td>%TRIM</td>
<td>This macro returns the argument passed to it without any trailing blanks in an unquoted form.</td>
</tr>
<tr>
<td>%VERIFY</td>
<td>This macro returns the position of the first character in the argument that is not in the target value.</td>
</tr>
</tbody>
</table>
To help illustrate a SASAUTOS macro, we will display the contents of the %TRIM autocall macro below. The purpose of the %TRIM autocall macro is to remove (or trim) trailing blanks from text and return the result.

%TRIM AUTOCALL Macro

```sas
%macro trim(value);
%*********************************************************************;
%* MACRO: TRIM
%*
%* USAGE: 1) %trim(argument)
%*       2) %trim
%* DESCRIPTION:
%*   This macro returns the argument passed to it without any trailing blanks in an unquoted form. The syntax for its use is similar to that of native macro functions.
%*   Eg. %let macvar=%trim(&argtext)
%* NOTES:
%*   None.
%*********************************************************************;
%local i;
%do i=%length(&value) %to 1 %by -1;
  %if %qsubstr(&value,&i,1)^=%str( ) %then %goto trimmed;
%end;
%trimmed: %if &i>0 %then %substr(&value,1,&i);
%mend;
```

Tip #4 – Compiling a Stored Macro with the Compiled Macro Facility

A macro can be compiled once and the compiled version stored so it can be used over and over again. This approach saves time and resources because the macro does not have to be compiled each time it is called. To take advantage of this time-saving approach, you will need to either verify and/or turn on the SAS System options: MSTORED and SASMSTORE. You will also need to specify the /STORE option of the %MACRO statement. It is worth mentioning that during macro compilation only macro statements are compiled, so be aware that non-macro text and macro references are not evaluated during the compilation phase – but during macro execution.

<table>
<thead>
<tr>
<th>SAS Option</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>MSTORED</td>
<td>Turns on the Compiled Macro Facility so you can take advantage of this feature.</td>
</tr>
<tr>
<td>SASMSTORE</td>
<td>Specifies the libref associated with the SAS catalog, SASMACR, of stored compiled macros.</td>
</tr>
</tbody>
</table>

Tip #5 – Streamlining Command-line DMS Commands with a Macro

The macro language is a wonderful tool for streamlining frequently entered SAS Display Manager System (DMS) commands to reduce the number of keystrokes. By embedding a series of DMS commands inside a simple macro, you’ll not only save by not having to enter them over and over again, but you’ll improve your productivity as well. The following macro code illustrates a series of DMS commands being strung together in lieu of entering them individually on a Display Manager command line. The commands display and expand the SAS Log to full size respectively, and then position the cursor at the top of the log. Once the macro is defined, it can be called by entering %POSTSUBMIT on any DMS command line to activate the commands.

Macro Code

```sas
%MACRO postsubmit;
  Log;
  Clear;
  Zoom;
%mend;
```
Tip #6 – Assigning a Defined Macro to a Function Key

To further reduce keystrokes and enhance user productivity even further, a call to a defined macro can be saved to a Function Key. The purpose for doing this would be to allow for one-button operation of any defined macro. To illustrate the process of saving a macro call to a Function Key, the %POSTSUBMIT macro defined in the previous tip is assigned to Function Key F12 in the KEYS window. The partial KEYS window is displayed to illustrate the process.

KEYS Window

<table>
<thead>
<tr>
<th>Key</th>
<th>Definition</th>
</tr>
</thead>
<tbody>
<tr>
<td>F1</td>
<td>help</td>
</tr>
<tr>
<td>F2</td>
<td>reshow</td>
</tr>
<tr>
<td>F3</td>
<td>end;</td>
</tr>
<tr>
<td>...</td>
<td>...</td>
</tr>
<tr>
<td>F10</td>
<td>keys</td>
</tr>
<tr>
<td>F11</td>
<td>command focus</td>
</tr>
<tr>
<td>F12</td>
<td>%POSTSUBMIT</td>
</tr>
</tbody>
</table>

Tip #7 – Defining Positional Parameters

Macros are frequently designed to allow the passing of one or more parameters. This allows the creation of macro variables so text strings can be passed into the macro. The order of macro variables as positional parameters is specified when the macro is coded. The assignment of values for each positional parameter is supplied at the time the macro is called. It is worth noting that while any number of positional parameters can be defined for a macro, there should be an established limit of no more than three parameters specified using a logical and natural order. Should there be a need to define more than three macro parameters, it is recommended that a macro with keyword parameters be defined instead.

To illustrate the definition of a two positional parameter macro, the following macro was created to display all table names (data sets) that contain the variable TITLE in the user-assigned MYDATA libref as a cross-reference listing. To retrieve the needed type of information, you could execute multiple PROC CONTENTS against selected tables. Or in a more efficient method, you could retrieve the information directly from the read-only Dictionary table COLUMNS with the selected columns LIBNAME, MEMNAME, NAME, TYPE and LENGTH, as shown. For more information about Dictionary tables, readers may want to view the “free” SAS Press Webinar by Kirk Paul Lafler at http://support.sas.com/publishing/bbu/webinar.html#lafler2 or the published paper by Kirk Paul Lafler, Exploring Dictionary Tables and SASHELP Views.

Macro Code

```sas
%MDEF COLUMNS(LIB, COLNAME);
PROC SQL;
   SELECT LIBNAME, MEMNAME, NAME, TYPE, LENGTH
   FROM DICTIONARY.COLUMNS
   WHERE UPCASE(LIBNAME)="&LIB" AND
       UPCASE(NAME)="&COLNAME" AND
       UPCASE(MEMTYPE)="DATA";
QUIT;
%MEND COLUMNS;
%COLUMNS(MYDATA,TITLE);
```
Now let's examine another useful macro that is designed with a positional parameter. The following macro is designed to accept one positional parameter called &LIB. When called, it accesses the read-only Dictionary table TABLES to display each table name and the number of observations in the user-assigned MYDATA libref. This macro provides a handy way to quickly determine the number of observations in one or all tables in a libref without having to execute multiple PROC CONTENTS by using the stored information in the Dictionary table TABLES.

Macro Code

%MACRO NUMROWS(LIB);
  PROC SQL;
    SELECT LIBNAME, MEMNAME, NOBS
    FROM DICTIONARY.TABLES
    WHERE UPCASE(LIBNAME)="&LIB" AND
      UPCASE(MEMTYPE)="DATA";
  QUIT;
%MEND NUMROWS;
%NUMROWS(MYDATA);

After Macro Resolution

PROC SQL;
  SELECT LIBNAME, MEMNAME, NOBS
  FROM DICTIONARY.TABLES
  WHERE UPCASE(LIBNAME)="MYDATA" AND
    UPCASE(MEMTYPE)="DATA";
QUIT;
Tip #8 – Referencing Macro Variables Indirectly
In each of the previous examples, a macro variable began with a single ampersand, for example, &macroname. When referenced, a macro variable defined this way is resolved using a direct approach by the macro facility to an assigned value. Although this represents the most common approach to defining and referencing a macro variable, it is not the only way a macro variable can be referenced. An alternate, and more dynamic approach supported by the macro facility is its ability to handle compound expressions consisting of a macro variable beginning with, as well as containing embedded ampersands, for example, &&TYPE&n.

Using indirect macro variable references, the next example illustrates a call to a macro containing an iterative %DO loop. The macro variable RATING1 through RATING5 contains the values G, PG, PG-13, PG-17, and R. To resolve the macro references in macro RATING, the macro processor first resolves the entire reference from left to right, resolving any pair of ampersands to a single ampersand followed by processing the next part of the reference. The macro processor then returns to the beginning of the preliminary result, resolving from left to right and continuing the process over again, as before, until all ampersands have been fully processed and the resulting macro variable produced.

Macro Code

```
%LET RATING1 = G;
%LET RATING2 = PG;
%LET RATING3 = PG-13;
%LET RATING4 = PG-17;
%LET RATING5 = R;
%MACRO RATING(STOP);
  %DO N=1 %TO &STOP;
    %PUT &RATING&N;
  %END;
%MEND RATING;

%RATING(3);
```

Output

G
PG
PG-13
Conclusion
The macro language provides SAS users with a powerful language environment for constructing a library of powerful tools, routines, and reusable programs. It offers a comprehensive set of statements, options, functions, and has its own compiler. Once written and debugged macro programs can be stored in a location on your operating environment that can be referenced and accessed using an autocall macro environment. Macros can also be compiled providing for a more efficient process for executing macros because the macro does not have to be compiled over and over again. Finally, users are able to design and construct reusable macro tools that can be used again and again.

References
Lafler, Kirk Paul (2009), “Building Reusable and Highly Effective Tools with the SAS® Macro Language,” PharmaSUG 2009 Conference, Software Intelligence Corporation, Spring Valley, CA, USA.
Lafler, Kirk Paul (2009), SAS System Macro Language Course Notes, Fifth Edition. Software Intelligence Corporation, Spring Valley, CA, USA.
Lafler, Kirk Paul (2008), Exploring DICTIONARY Tables and SASHELP Views, Software Intelligence Corporation, Spring Valley, CA, USA.
Lafler, Kirk Paul (2006), Exploring DICTIONARY Tables and SASHELP Views, Software Intelligence Corporation, Spring Valley, CA, USA.

Acknowledgments
The author would like to thank the SGF 2015 Hands-on Workshop Section Chair, for accepting my abstract and paper; Tyler Smith, the SGF 2015 Conference Chair; and the SGF 2015 Conference and Executive Committees for organizing a great conference!

Trademarks Citations
SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.
Author Information
Kirk Paul Lafler is consultant and founder of Software Intelligence Corporation and has been using SAS since 1979. He is a SAS Certified Professional, provider of IT consulting services, trainer to SAS users around the world, mentor, and sasCommunity.org emeritus Advisory Board member. As the author of six books including Google® Search Complete (Odyssey Press. 2014); PROC SQL: Beyond the Basics Using SAS, Second Edition (SAS Press. 2013); PROC SQL: Beyond the Basics Using SAS (SAS Press. 2004); Kirk has written more than five hundred papers and articles, been an Invited speaker and trainer at four hundred-plus SAS International, regional, special-interest, local, and in-house user group conferences and meetings, and is the recipient of 23 “Best” contributed paper, hands-on workshop (HOW), and poster awards.

Comments and suggestions can be sent to:

Kirk Paul Lafler
Senior SAS® Consultant, Application Developer, Data Scientist, Trainer and Author
Software Intelligence Corporation
E-mail: KirkLafler@cs.com
LinkedIn: http://www.linkedin.com/in/KirkPaulLafler
Twitter: @SASnerd